La gestion des erreurs

Les erreurs d'exécution se produisent lorsqu'une instruction tente d'accomplir une opération non valide.

Chaque erreur d'exécution provoque l'arrêt des macros. La ligne incriminée est surlignée en jaune et un message d'alerte s'affiche à l'écran.

Le numéro d'erreur ainsi qu'une description courte s'affichent dans la boîte de dialogue.

Le bouton Fin permet de sortir immédiatement de la procédure.

Le bouton Débogage ferme le message d'alerte et affiche la ligne qui a déclenché l'erreur.

Le bouton Aide ouvre l'aide associée au code d'erreur.

L'application dispose d'un gestionnaire d'erreurs pour prendre en compte ce genre de désagréments.

C'est l'instruction On Error qui valide la gestion d'erreur. Le gestionnaire est ensuite activé lorsque survient une erreur.

 On Error GoTo

Effectue un branchement vers une ligne spécifiée.

Par exemple

On Error goTo errorHandler

Cette instruction indique l'emplacement de la procédure qui gère les erreurs. Lorsqu'une erreur survient, l'exécution de la macro se déplace automatiquement jusqu'à cette ligne : errorHandler, dans l'exemple ci dessous.

On Error GoTo errorHandler doit être placé juste après la déclaration des variables, ou au moins avant la première ligne de procédure risquant de provoquer une erreur.

errorHandler: doit être placé en fin de procédure, éventuellement précédé par l'instruction Exit Sub.

Sub laMacro()

    'Placé en début de macro:

    'Si une erreur survient, on va à la ligne "errorHandler"

    On Error GoTo errorHandler

    '

    'La procédure

    '

    'Permet de sortir de la procédure et évite la gestion d'erreur (errorHandler), si la macro

    's'est déroulée sans encombre.

    Exit Sub

errorHandler:

    'indique le numéro et la description de l'erreur survenue

    MsgBox Err.Number & vbLf & Err.Description

End Sub

S'il n'y a pas de problème, la macro suit son cours et se termine par l'instruction Exit Sub. Dans le cas contraire, la macro va directement à la ligne errorHandler: pour gérer l'erreur.

Vous pouvez aussi ajouter  l'instruction Resume  si  vous  souhaitez  reprendre  l'exécution au niveau de  la  ligne à l'origine de l'erreur:

Sub UtilisationInstructionResume()

    Dim i As Integer, j As Integer

    On Error GoTo ErrorHandler

    'Par défaut i=0 (et va provoquer une erreur)

    j = 5 / i

    MsgBox j

    Exit Sub

ErrorHandler:

    MsgBox "Erreur: " & Err.Number & vbCrLf & Err.Description

    If i = 0 Then i = 1

    'Reprend l'exécution au niveau de la ligne à l'origine de l'erreur.

    Resume

End Sub
On Error GoTo 0

Invalide le gestionnaire d'erreurs précédemment créé par l'instruction On Error Goto.

Chaque erreur d'exécution provoque l'arrêt des macros, après la ligne On Error GoTo 0.

Dans l'exemple suivant, le gestionnaire d'erreur joue son rôle jusqu'à la ligne On Error GoTo 0. Ensuite la division par zéro fait planter la procédure.

Sub laMacro()

    Dim x As Double

    'Permet d'atteindre la ligne "errorHandler" si une erreur survient.

    On Error GoTo errorHandler

    MsgBox 5 / 2 'Pas de problème

    x = 5 / 0 'Division par zero = erreur

    MsgBox x

errorHandler:

    'Invalide le gestionnaire d'erreur.

    On Error GoTo 0

    'Provoque une erreur d'exécution car il n'y a plus de gestion.

    MsgBox 5 / 0

End Sub

On Error Resume Next

Permet de continuer la procédure et passe directement à la ligne suivante, en cas d'erreur.

Cette instruction est pratique pour contourner une ligne pouvant poser problème.

Par exemple, lorsque vous vérifiez l'existence d'une feuille dans le classeur:

Sub laMacro()

    Dim Ws As Worksheet

    'Evite le message d'erreur si la feuille n'existe pas.

    On Error Resume Next

    Set Ws = ThisWorkbook.Worksheets("NomFeuille")

    On Error GoTo 0

    '(Ws = Nothing quand l'objet attribué à la variable n'existe pas)

    If Not Ws Is Nothing Then

        MsgBox "La feuille existe dans le classeur."

        Else

        MsgBox "La feuille n'existe pas dans le classeur."

    End If

End Sub 

Cette  instruction doit être employée avec prudence  car elle masque  tous  les bugs qui peuvent  survenir dans  la procédure. Vous ne  recevrez aucune  information pour  localiser une erreur qui pourrait avoir une  influence sur  le résultat final de votre macro. Utilisez On Error Resume Next seulement si vous ne pouvez pas faire autrement, ou de façon ciblée sur une ligne précise (Comme dans l'exemple précédent).
Description de l'objet Err

L'objet Err contient des informations sur les erreurs d'exécution. Lorsqu'une erreur se produit, les propriétés de l'objet Err stockent d'une part des informations qui identifient l'erreur et d'autre part des informations permettant de gérer cette erreur.

Clear

La méthode Clear réinitialise le contenu de l'objet Err à la suite du traitement d'une erreur.

Les propriétés de l'objet Err sont alors remises à zéro ou remplacées par des chaînes de longueur nulle.

La méthode Clear est appelée automatiquement dès que l'une des instructions suivantes est exécutée:

* Tout type d'instruction Resume.

* Exit Sub, Exit Function ou Exit Property.

* Toute instruction On Error.

Si la méthode Clear n'était pas utilisée dans l'exemple suivant, le message s'afficherait à chaque tour de boucle après le déclenchement de la première erreur.

Sub laMacro()

    Dim i As Integer

    Dim Cible As Byte

    On Error Resume Next

    For i = 1 To 5

        'Renvoie une valeur aléatoire entre 1 et 400

        Randomize

        Cible = Int((400 * Rnd) + 1)

        'Une erreur survient si le résultat est supérieur à 255

        '(Les variables Byte acceptent uniquement des valeurs entre 0 et 255)

        If Err.Number <> 0 Then

            MsgBox "Erreur"

            'Réinitialise les propriétés de l'objet Err.

            Err.Clear

        End If

    Next i

End Sub
Description

La propriété Description  renvoie ou définit une chaîne de caractères qui correspond à une courte description de l'erreur.

Sub DescriptionErreur()

    Dim Obj As OLEObject

    On Error Resume Next

    'Déclenchement erreur:

    'Spécifie un objet qui n'existe pas dans la feuille

    Set Obj = Worksheets("Feuil1").Textbox1

    MsgBox Err.Description

End Sub

